3.3. Директивы моделиpования 





Программа PSpice рассчитывает следующие характеристики электронных цепей: 


(  режим цепи по постоянному току в “рабочей точке”; 


(  режим по постоянному току при вариации источников постоянного напряжения или тока, температуры и других параметров цепи; 


(  чувствительность характеристик цепи к вариации параметров компонентов в режиме по постоянному току; 


(  малосигнальные передаточные функции в режиме по постоянному току; 


(  характеристики линеаризованной цепи в частотной области при воздействии одного или нескольких сигналов; 


(  спектральная плотность внутреннего шума; 


(  переходные процессы при воздействии сигналов различной формы;


(  спектральный анализ; 


(  статистические испытания по методу Монте-Карло в любых режимах. 


Кроме того, производится многовариантный анализ любых характеристик при вариации температуры или любого параметра схемы.  С помощью модуля Optimizer выполняется параметрическая оптимизация (см. разд. 7.4).


Каждому виду расчета соответствует определенная директива. Большинство директив задается с помощью панели меню команды Analysis/Setup программы Schematics (рис. 2.16). Для таких директив  в скобках после их названия (см. ниже) указываются имена кнопок на этой панели. Остальные директивы вводят в задание на моделирование, размещая на схеме специальные символы из библиотеки special.slb (см. Приложение 1). Приведем описание всех директив, систематизируя их по разделам. С математической реализацией алгоритмов моделирования можно познакомиться в [1, 26].





3.3.1. Задание начальных условий 





Начальные значения узловых потенциалов по постоянному току задаются по директиве 





.IC   V(<номер узла>[,<номер узла>])=<значение ЭДС>* 





Приведем пример:





.IC V(5)=1.24  V(IN)=0 





К указанным узлам подключаются источники постоянного напряжения с внутренним сопротивлением 0,002 Ом, и рассчитывается режим по постоянному току. После завершения расчета эти источники отключаются ( так задаются начальные значения узловых потенциалов перед расчетом переходных процессов. 


Если в задании имеются и директива .NODESET, и директива .IC, то первая не будет выполняться при расчете режима по постоянному току перед началом анализа переходных процессов. 


Задание начального приближения узловых потенциалов по постоянному току производится по директиве 





.NODESET <V(<узел>[,<узел>])=<значение ЭДС>>* 





Приведем пример:





.NODESET V(9)=5.6 V(8,2)=4.95 





К указанным узлам подключаются источники постоянного напряжения с внутренним сопротивлением 0,002 Ом и выполняется одна итерация расчета режима по постоянному току. Затем эти источники автоматически отключаются и итерационный процесс расчета режима нелинейной схемы по постоянному току продолжается. Если заданные значения узловых потенциалов близки к точному решению, то процесс расчета завершается за меньшее количество итераций. Эта директива полезна при расчете очень больших схем по частям и расчете схем с несколькими устойчивыми состояниями. 


Запись в файл узловых потенциалов  (Save Bias Point) осуществляется по директиве





 .SAVEBIAS <"имя файла"> < OP |  DC | TRAN> [NOSUBCKT] 


+ [TIME=<значение> [REPEAT]] 


+ [TEMP=<значение>] [STEP=<значение>] 


+ [MCRUN=<значение>] 


+ [DC=<значение>] [DC1=<значение>] [DC2=<значение>] 





В файл с указанным именем записываются значения узловых потенциалов схемы для анализа указанного вида OP, DC или TRAN. Для каждого вида анализа в одном задании должна быть отдельная директива. 


Параметр NOSUBCKT запрещает запись в файл узловых потенциалов внутренних узлов макромоделей. 


Параметры TIME=<значение> [REPEAT] определяют моменты времени, в которые запоминаются узловые потенциалы схемы при расчете переходных процессов (режим TRAN). Ели параметр REPEAT не указан, то узловые потенциалы запоминаются в первый момент времени, равный указанному значению (с точностью до шага интегрирования) или превышающий его. При наличии параметра REPEAT параметр <значение> равен интервалу времени, с которым узловые потенциалы периодически запоминаются, при этом в файле сохраняется только последняя запись. 


Параметр TEMP=<значение> задает температуру, для которой сохраняются узловые потенциалы при вариации температуры, а необязательный параметр STEP=<значение> задает интервал температур, через который обновляется запись в файл. 


Параметр MCRUN=<значение> определяет количество вариантов расчетов, через которые обновляется запись в файл при статистическом анализе или расчете по методу наихудшего случая. 


Параметры DC=<значение>, DС1=<значение> и DC2=<значение> задают значения параметров, варьируемых по директиве .DC, для которых производится запись в файл. Параметр DC=<значение> используется только при вариации одной переменной, а DC1 и DC2 (  при вариации двух переменных (соответственно первой и второй). 


Образуемый на диске текстовый файл имеет следующий формат. Сначала на одной или более строках в виде комментариев указываются имя схемы, заголовок, дата создания, затраты времени на расчет, тип анализа, температура и т. п. Далее помещается единственная директива .NODESET, содержащая полную информацию об узловых потенциалах схемы. Поэтому после загрузки файла по директиве .LOADBIAS эти потенциалы будут автоматически установлены. 


Одно из типичных применений директив .SAVEBIAS и .LOADBIAS (  исследование больших схем, для которых расчет режима по постоянному току занимает значительное время. Тогда после расчета режима по постоянному току он может быть сохранен по директиве .SAVEBIAS и в дальнейшем перед выполнением других видов анализа (переходных процессов, частотных характеристик, статистического анализа и т. п.) узловые потенциалы в этом режиме могут быть загружены по директиве .LOADBIAS. Единственное ограничение (  директивы .SAVEBIAS и .LOADBIAS должны помещаться в разных заданиях на моделирование. Заметим, что при модификации схемы пользователь может для улучшения сходимости вручную отредактировать файл режима, добавляя или исключая из директивы .NODESET ряд узлов схемы и корректируя их потенциалы. 


Приведем примеры:





.SAVEBIAS   "DC_POINT.NOD"   OP 


.SAVEBIAS   "TR_DATA.TRN"   TRAN   NOSUBCKT   TIME=10us 


.SAVEBIAS   "DC.BSP"   DC   MCRUN=3   DC1=2.5   DC2=10 





Считывание из файла узловых потенциалов (Load Bias Point) производится по директиве





 .LOADBIAS <"имя файла"> 





Из файла, созданного в предыдущем сеансе работы с программой PSpice, считываются узловые потенциалы по постоянному току. Для выполнения расчета переходных процессов по директиве .TRAN с заданными начальными условиями этот файл можно предварительно отредактировать и заменить директиву .NODESET на .IC. Для передачи содержания файла узловых потенциалов, указанного в директиве .LOADBIAS, в выходной файл результатов (с расширением .out), необходимо в директиве .OPTIONS указать параметр EXPAND. 


Приведем примеры:





.LOADBIAS   "DC_POINT.NOD" 


.LOADBIAS   "D:\PSPICE\JOB\TR_DATA.TRN" 





3.3.2. Pасчет стандартных характеристик 





Расчет режима цепи по постоянному току (параметров “рабочей точки”). Режим по постоянному току всегда рассчитывается в начале моделирования перед выполнением других видов анализа без указания специальных директив. Результаты расчетов выводятся в текстовый файл .out в виде таблицы узловых потенциалов и списка токов независимых источников. Список малосигнальных параметров линеаризованных схем замещения полупроводниковых приборов и нелинейных управляемых источников выводится в выходной файл результатов при наличии директивы .OP (Bias Point Detail). 


Кроме того, анализ по постоянному току выполняется перед расчетом переходных процессов по директиве .TRAN для определения начальных условий (если отсутствует ключевое слово SKIPBP) и перед анализом в частотной области по директиве .AC для линеаризации нелинейных компонентов в окрестности режима по постоянному току. 


Обсудим особенности расчета режима по постоянному току. В программе он рассчитывается итерационным методом Ньютона( Рафсона. В отсутствие сходимости рекомендуется по директиве .OPTIONS увеличить максимальное количество итераций ITL1 (по умолчанию ITL1=40) [51]. Для повышения скорости сходимости рекомендуется с помощью директивы .NODESET устанавливать начальные значения узловых потенциалов, как можно более близкие к ожидаемому режиму по постоянному току (при отсутствии этой директивы все узловые потенциалы на начальной итерации полагаются равными нулю). Если решение методом Ньютона( Рафсона не сходится, программа автоматически переходит к методу вариации напряжений источников питания, который ценой увеличения затрат машинного времени обеспечивает сходимость решения в большинстве случаев. Приближенные значения режима по постоянному току с помощью директивы .NODESET обязательно указываются при анализе схем, имеющих несколько устойчивых состояний. 


Многовариантный расчет режима по постоянному току (DC Sweep). Расчет режима по постоянному току производится при вариации одного или нескольких источников постоянного напряжения или тока, температуры, параметров моделей компонентов схемы и глобальных параметров по директивам





.DC  [LIN] <имя 1-й переменной> <начальное значение> 


+ <конечное значение> <приращение> 


+ [<имя 2-й переменной> <начальное значение> 


+ <конечное значение> <приращение>]* 





.DC   <OCT | DEC> <имя 1-й переменной> <начальное значение> 


+ <конечное значение> <количество точек> 


+ [<имя 2-й переменной> <начальное значение> 


+ <конечное значение> <количество точек>]* 





.DC <имя 1-й переменной> LIST <значение>* 


+ [<имя 2-й переменной> LIST <значение> *]* 





Режим по постоянному току рассчитывается для нескольких значений варьируемых переменных, в качестве которых могут выступать: 


(  имена независимых источников напряжения или тока;


(  параметры моделей компонентов (указывается тип компонента, имя модели и в круглых скобках имя варьируемого параметра);


(  температуры (в качестве ее имени указывается ключевое слово TEMP);


(  глобальных параметров (указывается ключевое слово PARAM и вслед за ним имя варьируемого глобального параметра, определенного ранее). 


Характер изменения переменных задается ключевыми словами: 


LIN (  линейный масштаб (ключевое слово LIN можно не указывать); 


DEC, OCT (  логарифмический масштаб декадами или октавами; 


LIST (  список значений. 


Если указаны спецификации двух варьируемых параметров, то первый параметр изменяется в заданных пределах для каждого значения второго параметра. Такой вложенный цикл удобен, в частности, для построения статических характеристик полупроводниковых приборов. 


Максимальное количество итераций при переходе к следующему варианту по умолчанию устанавливается равным достаточно малой величине ITL2=20. Поэтому в схемах, чувствительных к вариациям параметров, могут возникнуть проблемы со сходимостью. В этих случаях рекомендуется по директиве .OPTIONS увеличить значение ITL2. 


Приведем примеры:





.DC   VIN  0.5  5.0  0.25 


.DC   LIN  VDS  0  10  .5  VGS  0  5  1 


.DC   VCE  0V  10V  .25V  IB  0  10MA  1MA 


.DC   RES   MODRES(R)  0.75  1.5  0.05 


.DC   DEC   NPN  KT315A(BF)  20  100  10 


.DC   TEMP   LIST  -50  0  27  60  80 


.DC   PARAM   VPOWER  7.5  12  .5 





Первый пример задает диапазон изменения напряжения источника VIN от 0,5 до 5,0 В с шагом 0,25 В. Второй и третий примеры демонстрируют вложенные циклы изменений двух источников. В четвертом и пятом примерах варьируются сопротивление резистора R и коэффициент передачи тока BF биполярного транзистора. В шестом варьируется температура, заданная списком значений. В седьмом (  глобальный параметр VPOWER, задающий ЭДС источника питания. 


В качестве еще одного примера покажем, как рассчитывают семейство выходных характеристик транзистора (рис. 3.15), используя вложенные циклы:





VС   1   0   DС   10V 


VB   2   0   DC   0.5V 


Q1   1   2   0   KT315A 


.MODEL   KT315A   NPN (IS= ...) 


.DC   VC   0V   10V   0.2V   VB   0V   1V   0.05V 


.PROBE 


�


Рис. 3.15. Схема измерений выходных характеристик транзистора





Результаты расчета режима цепи по постоянному току выводятся по директивам .PRINT, .PLOT или .PROBE. 


После завершения вариации параметров по директиве .DC варьируемые параметры принимают номинальные значения, которые они имели до многовариантного анализа. 


Замечание. При выполнении режима .DC при анализе цепей, обладающих регенеративными обратными связями (например, триггера Шмидта), не удается проанализировать точки неустойчивого равновесия. Для этого рекомендуется вместо режима .DC выполнить расчет переходных процессов .TRAN, введя источники медленно нарастающего и спадающего по линейному закону напряжения. 


Чувствительность в режиме малого сигнала (Sensitivity). Рассчитывается по директиве 





.SENS <выходная переменная>* 





Выходные переменные указываются по тому же формату, что и в директиве .PRINT для режима DC. При этом накладывается ограничение: если выходная переменная должна быть током, то допускается только ток через независимые источники напряжения. После линеаризации цепи в окрестности рабочей точки рассчитывается чувствительность каждой из указанных выходных переменных к изменению параметров всех компонентов и моделей. Поэтому объем результатов расчета чувствительностей может быть огромным. Результаты расчета выводятся в файл .out. 


Приведем пример. Если предположить, что цепь состоит из компонентов R1, R2, C1 и т. д., то по директиве


 


.SENS   V(9)   V(4,3)   I(VCC) 





будут рассчитаны чувствительности dV(9)/dR1, dV(9)/dR2, dV(9)/dC1, ..., dV(4,3)/dR1 ... 


Расчет малосигнальных передаточных функций (Transient Function). В режиме по постоянному току они рассчитываются по директиве 





.TF <выходная переменная> <имя источника напряжения или тока> 





после линеаризации цепи в окрестности рабочей точки. Выходные переменные имеют тот же формат, что и по директиве .PRINT. Если выходная переменная должна быть током, то это ток через независимый источник напряжения. Результаты расчетов выводятся в выходной файл .out без обращения к директивам .PRINT или .PLOT. Приведем примеры:





.TF   V(5)  VIN 


.TF   V(15, 14)   I(VDRIV) 





В первом случае рассчитывается передаточная функция dV(5)/dVIN, а во втором (  dV(15,14)/dI(VDRIV). Кроме того, всегда рассчитываются входные и выходные сопротивления. 


Расчет переходных процессов (Transient). Они рассчитываются по директиве


 


.TRAN[/OP] <шаг вывода данных> <конечное время> 


+ [<начальный момент времени вывода данных> [<максимальный шаг>]]  


+ [SKIPBP] 





Переходные процессы всегда рассчитываются с момента t=0 до момента <конечное время>. Перед началом расчета переходных процессов рассчитывается режим по постоянному току. Шаг интегрирования выбирается автоматически. Результаты вычислений выводятся в виде таблиц или графиков с интервалом времени, задаваемым параметром <шаг вывода данных> (применяется квадратичная интерполяция между дискретными отсчетами). Если задан параметр <начальный момент времени вывода данных>, то вывод результатов расчетов подавляется на интервале времени от t=0 до указанного значения. Максимальное значение шага интегрирования устанавливается параметром <максимальный шаг>; если он не указан, то максимальный шаг интегрирования устанавливается равным <конечное время>/50. 


Режим по постоянному току определяет начальные условия для расчета переходных процессов. Это связано с тем, что значения источников сигналов в момент t=0 могут отличаться от их постоянных составляющих. При этом в выходной файл .out выводятся только значения узловых потенциалов в режиме по постоянному току. Указание в директиве .TRAN суффикса /OP выводит в этот файл полную информацию о режиме по постоянному току (как по директиве .OP). 


Если в конце директивы .TRAN указать параметр SKIPBP, то расчет режима по постоянному току отменяется. При этом начальные значения напряжений на емкостях и токов через индуктивности указываются в опциях вида IC= ..., включенных в описания конденсаторов и индуктивностей, а начальные значения узловых потенциалов указываются в директиве .IC. 


Замечание. В версиях младше Design Center 6.2 вместо SKIPBP использовалось ключевое слово UIC (Use Initial Conditions).


При использовании директивы .PROBE совместно с директивой .TRAN следует иметь в виду, что программа Probe может вывести график, состоящий не более чем из 8000 отсчетов переменных (при этом отсчеты переменных, указанных в директиве .PROBE, берутся не с шагом вывода данных, а с внутренним шагом интегрирования). Однако программа Probe позволяет последовательно построить графики отдельных секций данных. Кроме того, можно подавить передачу данных программе Probe на начальном участке переходного процесса, указав параметр <начальный момент времени вывода данных>. 


При моделировании смешанных аналого-цифровых цепей шаги интегрирования в аналоговых и цифровых устройствах выбираются разными независимо друг от друга. Шаг интегрирования, указываемый для информации на экране программы PSpice, равен шагу интегрирования аналоговой части цепи. Шаг интегрирования цифровой части определяется значениями задержек в цифровых компонентах. Его минимальное значение определяется параметром DIGFREQ директивы .OPTIONS, оно равно 1/DIGFREQ. По умолчанию DIGFREQ=10 ГГц, но не более 1015/TSTOP. Задержками меньше половины шага интегрирования пренебрегают. 


При отсутствии сходимости расчета переходного процесса рекомендуется по директиве .OPTIONS увеличить максимальное количество итераций на одном временном шаге ITL4 (по умолчанию ITL4=10) и при необходимости общее количество итераций ITL5. По умолчанию ITL5=5000, установка ITL5=0 устраняет это ограничение. 


Допустимая относительная ошибка расчета токов и напряжений задается опцией RELTOL (по умолчанию � EMBED Equation.2  ���), а абсолютные ошибки токов, зарядов и напряжений (  опциями ABSTOL (по умолчанию � EMBED Equation.2  ��� А), CHGTOL (по умолчанию � EMBED Equation.2  ��� Кулон), VNTOL (по умолчанию � EMBED Equation.2  ��� В). Однако нельзя устанавливать абсолютные ошибки чрезмерно малыми. В частности, при анализе сильноточных или высоковольтных цепей задание абсолютных ошибок по умолчанию может привести к потере точности вычислений из-за ограниченности разрядной сетки ЭВМ. 


Приведем примеры: 





.TRAN  5ms  500ms 


.TRAN/OP  5ms  500ms  100ms  SKIPBP 


.TRAN  5ms  500ms  0ms  0.5ms 





Рассмотрим в качестве более конкретного примера особенности расчета переходных процессов в транзисторном мультивибраторе (рис. 3.16). В начале рассчитывается режим мультивибратора по постоянному току в соответствии со следующим заданием: 





Multivibrator 


R1  1  2  1k 


R2  5  0  750 


R3  1  3  100k 


R4  1  4  100k 


R5  1  6  1k 


R6  7  0  750 


C1  2  3  0.01uF 


C2  4  6  0.01uF 


Q1  2  4  5  KT312A 


Q2  6  3  7  KT312A 


VP  1  0  DC  10V 


.LIB  QRUS.LIB 


.PROBE 


.END 


�


Рис. 3.16.  Схема мультивибратора





В результате расчета в выходном файле c расширением .out создается таблица узловых потенциалов





V(1)=10,0000 B,  V(2)=6,3538 B,  V(3)=3,4889 B,  V(4)=3,4889 B, 


V(5)=2,7835 B,  V(6)=6,3538 B,  V(7)=2,7835 B. 





Если после расчета режима по постоянному току непосредственно перейти к расчету переходных процессов, то мультивибратор, находящийся в одном из состояний равновесия, не возбудится. Предложим два способа возбуждения математической модели мультивибратора. 


1. Перед началом переходного процесса с помощью директивы .IC задаются начальные значения узловых потенциалов и (или) с помощью параметра IC= задаются начальные значения напряжений на конденсаторах, отличные от значений в состоянии равновесия. В схеме на рис. 3.16 достаточно изменить потенциал узла 3 на 1% с помощью директивы 





.IC V(3)=3.46v 





и указать директиву расчета переходных процессов 





.TRAN  100us  1.2ms 





Заметим, что если с помощью директивы .IC и параметров IC в описаниях конденсаторов и индуктивностей полностью описан режим схемы по постоянному току, то перед расчетом переходного процесса этот режим рассчитывать не нужно (  этот расчет отменяется с помощью параметра SKIPBP в директиве .TRAN: 





.TRAN  100us  1.2ms  SKIPBP 





Кроме того, для получения текущей информации о поведении мультивибратора полезно указать директиву 





.WATCH  TRAN  V(2)  V(3)  V(6) 





Если автоколебания не возникли, то все узловые потенциалы будут сохранять постоянное значение, что будет видно на экране программы PSpice и позволит прервать моделирование. 


2. Имитируем включение напряжения питания, вводя источник напряжения в виде импульсной функции c линейным передним фронтом и длительностью, равной конечному времени анализа (тогда ее можно явно не указывать)





VP  1  0  PULSE (0v, 10v, 50us) 





Кроме того, для обеспечения самовозбуждения мультивибратора необходимо нарушить симметрию схемы, изменив, например, одно из сопротивлений на 1%





R1  1  2  1.01k 





Спектральный анализ (Transient/Fourier Analysis). Он проводится по директиве 





.FOUR <частота первой гармоники f� EMBED Equation.2  ���> [количество гармоник] 


+ <выходная переменная>* 





Спектральный анализ производится с помощью быстрого преобразования Фурье (БПФ) после завершения расчета переходного процесса (в задании должна иметься директива .TRAN). Имена переменных, спектр которых должен быть рассчитан, указываются в списке <выходная переменная>. В директиве .FOUR задается частота первой гармоники f1 и количество гармоник. Максимальное количество гармоник n=100 (в версиях до Design Center 5.2 (  10 гармоник). По умолчанию рассчитываются первые 9 гармоник. В программе рассчитываются амплитуды постоянной составляющей A� EMBED Equation.2  ��� и остальных n гармоник A� EMBED Equation.2  ���, A� EMBED Equation.2  ���,..., A� EMBED Equation.2  ���. Спектральному анализу подвергается участок реализации переходного процесса длительностью T = 1/f� EMBED Equation.2  ���  в конце интервала анализа (чтобы завершились переходные процессы). Результаты спектрального анализа выводятся в выходной файл .out в виде таблиц без указания директив .PRINT, .PLOT или .PROBE. Кроме того, рассчитывается коэффициент нелинейных искажений (в процентах) по формуле





K = � EMBED Equation.2  ���.





Приведем пример: 





.FOUR  10KHZ  15  V(5)  V(6,7)  I(VSENS3) 





Спецификация выходных переменных составляется по тем же правилам, что в директивах .PRINT или .PLOT. 


Заметим, что спектральный анализ производится также при обработке данных в графическом постпроцессоре Probe (см. разд. 7.2). Отличие состоит в том, что при этом спектральному анализу подвергается целиком весь график, изображенный на экране дисплея, или его часть. 


Расчет характеристик в частотной области (AC Sweep). Они рассчитываются по директиве 





.AC  [ LIN ] [ OCT ]  [ DEC ] <n> <начальная частота> <конечная частота> 





Эта директива задает диапазон частот в пределах <начальная частота> ... <конечная частота>. Параметр LIN устанавливает линейный шаг по частоте, при этом n ( общее количество точек по частоте. Параметры OCT и DEC устанавливают логарифмический характер изменения частоты октавами и декадами соответственно. Параметр n определяет в таком случае количество точек по частоте на одной октаве или декаде. Расчет характеристик в частотной области производится после определения режима по постоянному току и линеаризации нелинейных компонентов (это делается автоматически, никаких дополнительных директив не требуется). Амплитуды и фазы одного или нескольких гармонических сигналов указываются при описании параметров независимых источников напряжения V или тока I. Результаты расчета комплексных амплитуд узловых напряжений и токов ветвей выводятся по директивам .PRINT, .PLOT или .PROBE. 


Анализ уровня внутреннего шума (AC Sweep/Noise Analysis). Анализ производится по директиве 





.NOISE  V(<узел>[,<узел>]) <имя> <n> 





Директива .NOISE указывается совместно с директивой .AC, в которой задается диапазон частот анализа. Источниками шума служат резисторы, ключи и полупроводниковые приборы, шумовые схемы замещения которых приведены в гл. 4. На каждой частоте f рассчитывается спектральная плотность выходного напряжения � EMBED Equation.2  ���, В� EMBED Equation.2  ���/Гц, обусловленная наличием статистически независимых источников внутреннего шума. Точки съема выходного напряжения указываются по спецификации V(<узел>[,<узел>]). К входным зажимам цепи подключается независимый источник напряжения или тока, <имя> которого приводится в списке параметров директивы .NOISE. Этот источник не является источником реального сигнала, он служит лишь для обозначения входных зажимов цепи, к которым пересчитывается выходной шум. Если ко входу подключается источник напряжения, то на входе рассчитывается эквивалентная спектральная плотность напряжения шума � EMBED Equation.2  ���, В� EMBED Equation.2  ���/Гц; если ко входу подключен источник тока, то рассчитывается эквивалентная спектральная плотность тока  � EMBED Equation.2  ���, А� EMBED Equation.2  ���/Гц. Уровень шума пересчитывается с выхода на вход делением спектральной плотности выходного напряжения � EMBED Equation.2  ��� на квадрат модуля соответствующей передаточной функции. Заметим, что внутреннее сопротивление реального генератора сигнала R� EMBED Equation.2  ��� должно быть включено в описание цепи как отдельный резистор. 


Если указан целочисленный параметр <n>, то на каждой n-й частоте в диапазоне анализа будет рассчитываться не только спектральная плотность суммарного шума, но и вклад в нее каждого шумового источника. Если параметр <n> не указан, то этот расчет не производится. 





Приведем примеры





.NOISE  V(5)  VIN 


.NOISE  V(101)  VSRC  20 


.NOISE  V(4, 5)  ISRC 





Результаты расчета уровней шума выводятся в выходной файл .out по директиве . PRINT или .PLOT





.PRINT  NOISE  <выходная переменная>* 


.PLOT   NOISE  <выходная переменная>* 





В качестве выходных переменных используются следующие имена: 


INOISE, DB(INOISE) ( � EMBED Equation.2  ��� в относительных единицах и децибелах; 


ONOISE, DB(ONOISE) ( � EMBED Equation.2  ��� в относительных единицах и децибелах. 


Например





.PRINT  NOISE  INOISE  ONOISE  DB(INOISE) 


.PRINT  NOISE  INOISE  ONOISE 


.PRINT  NOISE  ONOISE  DB(ONOISE) 





Графики спектральных плотностей можно построить с помощью программы Probe при наличии директивы .PROBE. Причем при наличии в этой директиве списка переменных имена переменных INOISE и ONOISE указывать не надо ( они передаются в файл результатов .dat   всегда. 


По результатам расчета спектральной плотности внутреннего шума легко вычисляется дифференциальный коэффициент шума линейного четырехполюсника, изображенного на рис. 3.17, а. Как известно [13], дифференциальный коэффициент шума равен 





� EMBED Equation.2  ���, 





где � EMBED Equation.2  ��� ( спектральная плотность напряжения шума, обусловленная шумом выходного сопротивления генератора R� EMBED Equation.2  ��� и внутренним шумом четырехполюсника, пересчитанная на его вход, � EMBED Equation.2  ���=INOISE� EMBED Equation.2  ���;  � EMBED Equation.2  ��� ( спектральная плотность напряжения шума выходного сопротивления генератора; � EMBED Equation.2  ��� Дж/� EMBED Equation.2  ��� ( постоянная Больцмана;  � EMBED Equation.2  ��� ( номинальная абсолютная температура. 


�


Рис. 3.17. Измерение коэффициента шума четырехполюсника при подключении ко входу источника напряжения (а) или тока (б)


Формула для расчета дифференциального коэффициента шума приобретает, таким образом, вид 





� EMBED Equation.2  ���





где сопротивление � EMBED Equation.2  ��� указывается в омах. Приведем фрагмент задания на расчет коэффициента шума четырехполюсника, изображенного на рис. 3.17, а: 





.TEMP  80 


VG  1  0  AC  1 


RG  1  2  75 


RLOAD  3  4  500 


CLOAD  3  4  30pF 


{описание четырехполюсника} 


.AC  LIN  21  0  1000HZ 


.NOISE  V(3,4)  VG 


.PRINT  INOISE  ONOISE 





Аналогично измеряется дифференциальный коэффициент шума при подключении на вход четырехполюсника источника тока (рис. 3.17, б): 





� EMBED Equation.2  ���





Расчет среднеквадратического отклонения выходного напряжения шума производится по формуле 





� EMBED Equation.2  ���





с помощью функции интегрирования s(x) программы Probe (разд. 7.2)





SQRT(S(V(ONOISE)*V(ONOISE))) 





Среднеквадратическое отклонение шума � EMBED Equation.2  ���  равно значению этой функции на верхней границе диапазона частот. 


При измерении коэффициента шума сопротивлению генератора � EMBED Equation.2  ��� должна быть приписана номинальная температура � EMBED Equation.2  ���, четырехполюснику ( его физическая температура, а сопротивлению нагрузки � EMBED Equation.2  ��� ( температура абсолютного нуля, так как его шумы обычно учитываются при расчете коэффициента шума последующего каскада. Назначение резисторам индивидуальных значений температуры производится с помощью параметра T_ABS.


В некоторых задачах могут потребоваться независимые источники шума. Они могут быть представлены в виде зависимых источников, управляемых током шумящего резистора (в связи с особенностями программы PSpice допускается управлять током источника ЭДС, включенного последовательно с шумящим резистором). 


На рис. 3.18, а представлена модель независимого источника шумового напряжения. Сопротивление шумящего резистора этого источника связано с требуемой спектральной плотностью напряжения шума � EMBED Equation.2  ��� соотношением � EMBED Equation.2  ��� Например, при � EMBED Equation.2  ��� B� EMBED Equation.2  ���/Гц описание модели на рис. 3.18, а имеет вид 





RN  1  0  60.4 


VN  1  0  DC  0 


HN  2  3  VN  1 


�


Рис. 3.18. Независимый источник напряжения (а) и тока (б) широкополосного шума





Модель независимого источника шумового тока представлена на рис. 3.18, б. Для нее сопротивление шумящего резистора связано со спектральной плотностью шумового тока � EMBED Equation.2  ��� соотношением RN=4kT/S� EMBED Equation.2  ���. Например, при � EMBED Equation.2  ��� A� EMBED Equation.2  ���/Гц модель описывается следующим образом: 





RN  1  0  16.56K 


VN  1  0  DC  0 


FN  2  3  VN  1 





3.3.3. Многовариантный анализ 





Вариация параметров (Parametric) назначается по директиве .STEP, имеющей следующие разновидности: 





.STEP [ LIN ] <имя варьируемого параметра> <начальное значение> 


+ <конечное значение> <шаг приращения параметра> 





.STEP [ OCT ] [ DEC ] <имя варьируемого параметра> 


+ <начальное значение> <конечное значение> <количество точек> 





.STEP <имя варьируемого параметра> LIST <значение>* 





На каждом шаге вариации параметров по очереди выполняются все виды анализа характеристик цепи, задаваемых директивами .DC, .AC, .TRAN и др. Варьироваться могут все параметры всех моделей компонентов и глобальные параметры за исключением: 


( параметров L и W МОП-транзистора (разрешается варьировать аналогичные параметры LD и WD); 


( температурных коэффициентов TC1, TC2 резисторов и других компонентов. 


Приведем примеры





.STEP  VIN  -.8  .8  .2 


.STEP  LIN  I2  5mA  -2mA  -0.1mA 


.STEP  RES  RMOD(R)  0.9  1.1  0.05 


.STEP  TEMP  LIST  0  20  27  50  80 


.STEP  PARAM  VPOWER  4  6  0.2 





Ключевое слово PARAM в последнем примере указывает, что после него следует имя глобального параметра, определенного ранее по директиве .PARAM. 


Дадим пояснения, как с помощью директивы .STEP организовать многовариантный анализ. 


Например, многовариантный анализ переходных процессов при изменении амплитуды A гармонического сигнала реализуется следующим образом: 





.PARAM  A=0 


VSIGNAL  1  0  SIN(0  {A} 1kHz) 


.STEP  PARAM  A  LIST  1  2  5  10 


.TRAN 0.1ms 5ms 





Обратим внимание, что при вариации глобальных параметров их необходимо предварительно объявить по директиве .PARAM. 


Изменение сопротивления резистора (и параметров других пассивных компонентов) осуществляется двояко. Во-первых, с помощью глобального параметра





.PARAM  P=1 


R1  2  0  {P} 


.STEP  PARAM  P  15,45,10 





Во-вторых, используя модель резистора





.MODEL  RMOD  RES(R=15) 


R1  2  0  RMOD  1 


.STEP  RES  RMOD(R)  15,45,10 





Здесь RMOD ( имя модели резистора, RES ( тип модели, R ( имя варьируемого параметра. 


В связи с тем что многовариантный анализ производится также с помощью директив .TEMP, .MC, .WCASE и .DC, в одном задании на моделирование вместе с директивой .STEP разрешается помещать только одну из них. Две директивы .STEP в одном задании не допускаются. 


Вариация температуры  (Temperature) производится по директиве 





.TEMP <температура>* 





Здесь указывается список значений температуры (по шкале Цельсия), для которых следует выполнить все указанные в задании директивы анализа характеристик. Если указано несколько значений температуры, то все виды анализа проводятся для каждой температуры. Если директива .TEMP не приведена, а в директиве .OPTIONS не указано другого значения температуры, то расчеты проводятся для номинальной температуры Tnom = 27 � EMBED Equation.2  ���C. 





3.3.4. Статистический анализ 





Статистический анализ по методу Монте-Карло (Monte Carlo). Он производится при статистическом разбросе параметров, описанных по директиве .MODEL. Случайное значение параметра x рассчитывается по формуле 





� EMBED Equation.2  ���





где � EMBED Equation.2  ��� ( номинальное значение параметра, указанное в директиве .MODEL; � EMBED Equation.2  ��� ( относительный разброс параметра x; � EMBED Equation.2  ��� ( центрированная случайная величина, принимающая значения на отрезке ((1, +1).  


Случайные величины x создаются с помощью генераторов случайных чисел. Величина относительного разброса каждого параметра � EMBED Equation.2  ��� и закон распределения случайной величины � EMBED Equation.2  ��� задаются опцией <спецификация случайного разброса параметра> директивы .MODEL, которая имеет вид





[DEV[/<генератор #>][/<закон распределения>] <разброс>[%]] 


[LOT[/<генератор #>][/<закон распределения>] <разброс>[%]] 





Параметр <генератор #> указывает номер генератора случайных чисел (от 0 до 9). С его помощью создаются коррелированные параметры. Случайные параметры, для которых не указаны номера генераторов случайных чисел, образуются с помощью индивидуальных независимых генераторов; они, естественно, являются некоррелированными. Для расчета значений разбросов параметров DEV и LOT используются различные генераторы, имеется 10 генераторов для параметров DEV и столько же для параметров LOT. 


В простейшем случае номера генераторов случайных чисел не указываются. Тогда параметры, имеющие опции DEV, получают независимые, а параметры, имеющие опции LOT, ( коррелированные случайные значения (последнее характерно для имитации разброса параметров партий изделий). 


Приведем примеры





.MODEL  RLOAD  RES  (R=1  DEV/GAUSS  5%  LOT/UNIFORM  10%) 


.MODEL  CMOD  CAP  (C=1  DEV/4/GAUSS  1% 


+   TC1  DEV/4/USER1  10%  LOT/5  2%) 





В программе имеются генераторы случайных величин с двумя стандартными законами распределения: 


UNIFORM ( равновероятное распределение на отрезке ((1, +1); 


GAUSS ( гауссовое распределение на отрезке ((1, +1) с нулевым средним значением и среднеквадратическим отклонением � EMBED Equation.2  ���=0,25 (т. е. создается усеченное гауссовое распределение на интервале (4� EMBED Equation.2  ���). 


Кроме того, пользователь может задать нестандартный закон распределения случайных величин � EMBED Equation.2  ��� с помощью директивы 





.DISTRIBUTION <имя> <<� EMBED Equation.2  ���> <P>>* 





Здесь параметр <имя> назначает имя закону распределения, который задается в табличной форме. Пары чисел задают значения случайной величины � EMBED Equation.2  ��� и соответствующую вероятность P. Всего может быть задано до 100 точек. Все значения � EMBED Equation.2  ��� должны находиться на интервале ((1, +1). Координаты точек должны указываться в порядке возрастания � EMBED Equation.2  ��� (допускается повторять предыдущие значения � EMBED Equation.2  ���). Между соседними точками производится линейная интерполяция. 


Приведем пример задания бимодального распределения, график которого изображен на рис. 3.19





.DISTRIBUTION  BI_MODAL  (-1,1)  (-.5,1)  (-.5,0)  (.5,0)  (.5,1)  (1,1) 


�


Рис. 3.19. Бимодальный закон распределения





Случайным параметрам, закон распределения которых не задан явно в директиве .MODEL, по умолчанию назначается распределение, указанное в опции DISTRIBUTION директивы .OPTIONS. 


Статистические испытания по методу Монте-Карло проводятся при расчете режима по постоянному току, переходных процессов или частотных характеристик по директиве 





.MC <n> [ DC ] [ TRAN ] [ AC ] <имя выходной переменной> 


+ <обработка результатов> [ LIST ] 


+ [ OUTPUT <cпецификация>] [ RANGE(<минимум>,<максимум>) ] 


+ [ SEED=<значение>] 





Параметр <n> задает количество статистических испытаний. Ключевые слова DC, TRAN, AC указывают вид анализа. После них указывается <имя выходной переменной>, подлежащей статистической обработке. 


При статистическом анализе предусматривается разнообразная статистическая обработка результатов моделирования, характер которой определяется с помощью опции <обработка результатов>, принимающей одно из следующих значений: 


YMAX ( расчет максимального отклонения текущей реализации от номинальной; 


MAX ( расчет максимального значения в каждой реализации; 


MIN ( расчет минимального значения в каждой реализации; 


RISE_EDGE(<значение>) ( определение момента первого пересечения заданного уровня снизу вверх (значение уровня задается в круглых скобках; в начале расчета значение реализации должно быть меньше этого уровня); 


FALL_EDGE(<значение>) ( определение момента первого пересечения заданного уровня сверху вниз (значение уровня задается в круглых скобках; в начале расчета значение реализации должно быть больше этого уровня). 


По необязательному ключевому слову LIST на печать выводится список значений всех случайных параметров во всех реализациях. 


В отсутствие ключевого слова OUTPUT характеристики цепи, указанные в директивах .PRINT, .PLOT или .PROBE, выводятся на печать или передаются в постпроцессор Probe один раз для номинального значения случайных параметров. С помощью ключевого слова OUTPUT их можно вывести требуемое число раз, задавая после этого слова следующие параметры: 


ALL ( во всех реализациях; 


FIRST <m> ( для первых m реализаций; 


EVERY <m> ( на каждой m-й реализации; 


RUNS <m� EMBED Equation.2  ���>,<m� EMBED Equation.2  ���>... ( для реализаций с указанными номерами m� EMBED Equation.2  ���, m� EMBED Equation.2  ���... 


После ключевого слова RANGE определяется диапазон значений, в пределах которого статистически обрабатывается выходная переменная. Если вместо минимального или максимального значения этого диапазона указать звездочку “*”, то граница диапазона примет значение -� EMBED Equation.2  ��� или +� EMBED Equation.2  ���. 


Переменная SEED задает начальное значение датчика случайных чисел. Оно может принимать нечетные значения в диапазоне 1...32 767 [6]. По умолчанию SEED=17 533. При одинаковых значениях SEED результаты статистического моделирования при последующих запусках программы повторяются. 


Приведем примеры:





.MC   5   TRAN  V(5)  YMAX   RANGE(-1,*) 


.MC   100  AC  VM(3)  YMAX   LIST  OUTPUT=ALL 





В старых версиях программы PSpice количество испытаний n ограничено 2000 при выводе данных в текстовый файл .out и 100 при наличии директивы .PROBE.   В Design Center 6.2 ( не более 1000 испытаний.


Расчет чувствительности и наихудшего случая (Worst Case). Для этого применяется директива 





.WCASE  [ DC ] [ TRAN ] [ AC ] <имя выходной переменной> 


+ <обработка результатов> [<опции>]* 





Виды анализа и параметры <имя выходной переменной>, <обработка результатов> такие же, как у директивы .MC. 


Проводятся расчеты характеристик цепи при вариации параметров, имеющих опции DEV или LOT. Сначала по очереди изменяются все указанные параметры, что позволяет оценить параметрическую чувствительность характеристик. Измененные значения параметров рассчитываются по формуле 





Новое значение = номинальное значение ( (1 + RELTOL), 





если не указана опция BY (см. ниже). В заключение рассчитываются характеристики цепи при одновременном изменении всех параметров по методу наихудшего случая. При этом значения параметров определяются по формуле 





Новое значение = номинальное значение  ( (1 ( � EMBED Equation.2  ���), 





Опции имеют следующий смысл. 


Ключевое слово OUTPUT ALL разрешает передавать в выходной файл характеристики цепи, указанные в директивах .PRINT, .PLOT и .PROBE, при всех комбинациях параметров при расчете чувствительности. В его отсутствие передаются данные лишь о номинальном режиме и наихудшем случае. 


После ключевого слова RANGE(<минимум>,<максимум>) указывается диапазон значений, в пределах которого производится статистическая обработка выходной переменной. Если вместо минимального или максимального значения этого диапазона указать звездочку “*”, то граница диапазона принимает значение  (� EMBED Equation.2  ��� или +� EMBED Equation.2  ���. 


Ключевые слова HI и LOW задают направление изменения параметров компонентов относительно номинальных значений при расчете наихудшего случая. Если определяется функция YMAX или MAX, по умолчанию назначается ключевое слово HI (положительное приращение), в противном случае ( LOW (отрицательное приращение). 


Ключевые слова VARY DEV, VARY LOT, VARY BOTH определяют характер случайного разброса параметров. По умолчанию назначается параметр VARY BOTH и случайный разброс задается обоими параметрами DEV и LOT; два других ключевых слова означают случайный разброс, определяемый только одним параметром DEV или LOT соответственно. 


Ключевые слова BY RELTOL, BYE <значение> задают относительное изменение параметров при расчете чувствительности. По умолчанию их изменение равно значению параметра RELTOL, указанному в директиве .OPTIONS. 


Ключевое слово DEVICES(<список типов компонентов>) назначает типы компонентов, параметры которых изменяются при расчете чувствительности и наихудшего случая. Список типов компонентов состоит из первых символов их имен, перечисляемых слитно, без пробелов. Например, вариация параметров только резисторов (R), конденсаторов (C) и биполярных транзисторов (Q) назначается с помощью ключевого слова DEVICES RCQ. 


Приведем примеры директивы .WCASE 





.WCASE  DC  V(4,5)  YMAX 


.WCASE  TRAN  V(1)  FALL_EDGE(3.5V)  VARY  BOTH  BY  RELTOL  DEVICES  RL 





В задании на моделирование может быть либо директива .MC, либо .WCASE, но не обе вместе. Так моделируется наихудший случай для чисто аналоговых устройств. 


Расчет наихудшего случая для смешанных устройств. При исследовании цифровых устройств наиболее актуально оценить влияние случайного разброса временных задержек цифровых ИС на характеристики устройства в целом. Сначала исследуются характеристики при номинальных значениях задержек, а затем интересно проверить работоспособность устройства при вариации задержек в заданных пределах. В версии PSpice 4 директива .WCASE для смешанных аналого-цифровых устройств неприменима, и поэтому приходилось одновременно всем задержкам присваивать минимальные, номинальные или максимальные значения. Эта информация полезна, но недостаточна, так как неприятности могут возникать не только при синхронных отклонениях задержек, но и при определенном сочетании максимальных отклонений отдельных варьируемых задержек. Эту возможность представляет директива .WCASE для Design Center  5.1. 


Расчет наихудшего случая при моделировании смешанных аналого-цифровых устройств отличается от моделирования чисто аналоговых устройств. При моделировании по методу наихудшего случая аналого-цифровых устройств аналоговые секции моделируются при номинальных значениях параметров. В цифровых компонентах варьируются задержки, имеющие в моделях динамики ненулевые минимальные и максимальные значения (см. разд. 6.2). При подаче на вход цифрового компонента воздействия выходной узел имеет неопределенное логическое состояние на интервале времени, равном разности между максимальной и минимальной задержкой. После прохождения сигнала через второй цифровой компонент длительность неопределенного состояния увеличивается и т. д. Это так называемое моделирование “с нарастающей неопределенностью”. 





3.3.5. Управление выдачей результатов 





Результаты расчетов в виде таблиц выводятся в выходной файл с расширением .out по директиве 





.PRINT[/DGTLCHG] [ DC ] [ TRAN ] [ AC ] [ NOISE ] <выходная переменная>* 





В одной директиве .PRINT можно выбрать только один вид анализа и привести список не более восьми выходных переменных. Одновременно в задании на моделирование можно поместить несколько таких директив. В таблицах каждая колонка соответствует одной переменной. В первой колонке помещается независимая переменная: постоянное напряжение (режим DC), время (режим TRAN) или частота (режим AC). Количество значащих цифр данных и максимальное количество строк в таблице определяются опциями NUMDGT и LIMPTS директивы .OPTIONS. Формат выходных переменных описан ниже. Приведем примеры:





.PRINT  DC  V(3)  V(2,3)  I(VIN) 


.PRINT  AC  VM(2)  VP(2)  VDB(5)  IR(6) 


.PRINT  TRAN  V(1)  V(R1)  V([RESET]) 





В последнем примере нецифровое имя узла RESET, начинающееся с буквы, заключено в квадратные скобки, чтобы отличить его от имени компонента. 


Логические состояния цифровых компонентов выводятся на внешние устройства после окончания моделирования обычным образом. В общем случае спецификация <выходная переменная> цифровых узлов имеет вид 


D(<имя узла>) 


Однако при наличии параметра /DGTLCHG префикс D можно опустить. Поэтому следующие две директивы вывода на печать логических состояний цифровых узлов SET и Q1 эквивалентны





.PRINT  TRAN  D(SET)  D(Q1) 


.PRINT/DGTLCHG  TRAN  SET  Q1 





Различие этих директив заключается в том, что по директиве .PRINT на печать выводятся состояния как цифровых, так и аналоговых узлов, а по директиве .PRINT/DGTLCHG ( только цифровых. 


Результаты в виде графиков выводятся в выходной файл по директиве 





.PLOT  [DC] [AC] [NOISE] [TRAN] <выходная переменная>* 


+ [(<нижняя граница>,<верхняя граница>)]* 





Смысл параметров такой же, что и в директиве .PRINT. Графики выводятся с помощью буквенно-цифровых символов независимо от типа печатающего устройства ( как на АЦПУ старых ЭВМ, для которых была создана программа SPICE, поэтому директивой .PLOT пользоваться на современных ПК не имеет смысла. На одном графике помещается до восьми кривых, однако количество директив .PLOT в одном задании не ограничено. Диапазон по оси x указан в директиве, устанавливающей вид анализа, а диапазон по оси y определяется с помощью параметров <нижняя граница>, <верхняя граница> или автоматически. Приведем примеры: 





.PLOT  DC  V(3)  V(2,3)  V(R1)  I(VIN) 


.PLOT  AC  VM(2)  VP(2)  VM(3,4)  VG(5) 


.PLOT  NOISE  INOISE  ONOISE 


.PLOT  TRAN  V(3)  V(2,3)  (0,5V)  ID(M2)  (-50mA, 50mA) 





Графический постпроцессор Probe подключается директивой 





.PROBE[/CSDF] [<выходная переменная>*] 





Если список выходных переменных не указан, то в файл результатов с расширением имени .dat заносятся потенциалы всех узлов цепи и токи всех компонентов, разрешенных для помещения в список выходных переменных (см. ниже). Обратим внимание на то, что при этом файл результатов может иметь огромные размеры и не поместиться в ОЗУ. Указание конкретного списка выходных переменных, передаваемых в программу Probe через файл с расширением .dat, сокращает размер этого файла. Приведем примеры





.PROBE 


.PROBE V(3) V(2,3) VM(2) VP(2) 





Обратим внимание, что в файл данных .dat всегда помещаются уровни внутреннего шума INOISE, ONOISE и данные о кривых гистерезиса магнитных сердечников B(H), поэтому при наличии списка выходных переменных их в него включать не надо. 


Максимальное количество отсчетов, передаваемых в постпроцессор Probe, равно 16 000 (при расчете переходных процессов это общее количество внутренних шагов, а не количество точек, выводимых на печать по директивам .PRINT и .PLOT). 


Обратим внимание, что по  директиве .PROBE в файл данных всегда передаются данные о спектральной плотности внутреннего шума INOISE, ONOISE и  кривых гистерезиса B(H) магнитных сердечников, даже если при наличии списка переменных они в него не включены.


По директиве .PROBE/CSDF создается файл результатов в текстовом виде с расширением имени .txt, который можно использовать для обмена данными с ЭВМ других типов  и для сопряжения с программами дополнительной обработки результатов. 


Замечание. Если в управляющей оболочке Schematic в меню Analysis/Probe Setup активизирована опция Automatically Run Probe After Simulation, то результаты расчетов автоматически передаются в программу Probe после завершения работы PSpice. При выборе опции Monitor Waveforms вывод графиков на экран программы Probe начинается одновременно с началом моделирования. 


Запись в файл результатов моделирования цифровых устройств производится по директиве (введена в версии Design Center 6.1)





.VECTOR  <количество узлов> <номер узла>*


+  [POS=<позиция столбца в таблице>  [FILE=<имя файла>]


+  [RADIX=“Binary”|”Hex”|”Octal”


+  [BIT=<номер бита>]]


+  [ SIGNAMES=<имена сигналов> ]





В файл заносятся значения моментов времени и логические уровни сигналов в перечисленных узлах. Формат файла такой же, какой применяется при создании файла цифрового генератора FSTIM.  Приведем примеры:





.VECTOR  1  CLOCK  SIGNAMES=SYSCLK


.VECTOR  4  DATA3  DATA2  DATA1  DATA0


.VECTOR  1  ADDR3  POS=2  RADIX=H  BIT=4


.VECTOR  1  ADDR2  POS=2  RADIX=H  BIT=3


.VECTOR  1  ADDR1  POS=2  RADIX=H  BIT=2


.VECTOR  1  ADDR0  POS=2  RADIX=H  BIT=1





По умолчанию создается файл, имеющий имя текущей схемы с расширением .vec.


Выдача текущих результатов анализа производится по директиве 





.WATCH [ DC ] [ AC ] [ TRAN ] [<выходная переменная> 


+ [<нижний предел>,<верхний предел>] ]* 





В процессе работы программы PSpice на нижней строке экрана в текстовой форме выводятся значения до 3-х выходных переменных типа V или I (арифметические выражения недопустимы). Каждая переменная может иметь свои пределы. Если ее текущее значение выходит за эти пределы ( раздается звуковой сигнал, после чего пользователь может прекратить расчеты или не обращать на это внимание. В последнем случае программа больше не будет проверять нахождение переменных в указанных пределах. Для каждого вида анализа DC, AC, TRAN записывается отдельная директива .WATCH. При этом недоступен вывод группового времени запаздывания (в режиме AC) и состояний цифровых узлов. 


Длина строки выходного файла устанавливается директивой


 


.WIDTH OUT = <значение> 





Параметр <значение> устанавливает количество колонок в выходном файле: 80 (по умолчанию) или 132. 


Выходные переменные. Опишем, как составляются имена выходных переменных, используемых в директивах .PRINT, .PLOT и .PROBE. В разных видах анализа имена несколько различаются. 


1. Режимы DC и TRAN. При расчете режима по постоянному току и переходных процессов используются выходные переменные, приведенные в табл. 3.17.





Таблица 3.17 





Общая форма �
Пример �
Пояснение �
�
V(<узел>) �
V(2) �
Потенциал узла �
�
V(<+узел>,<(узел>) �
V(2,4) �
Разность потенциалов узла �
�
V(<имя>) �
V(R1) �
Разность потенциалов между выводами двухполюсного компонента �
�
Vx(<имя>) �
VB(Q1) �
Потенциал вывода многополюсного компонента �
�
Vxy(<имя>) �
VCE(Q3) �
Разность потенциалов между выводами многополюсного компонента �
�
Vz(<имя>) �
VA(T2) �
Напряжение на входе или выходе линии передачи �
�
I(<имя>) �
I(D5) �
Ток через двухполюсный компонент �
�
Ix(<имя>) �
IB(Q1) �
Ток через указанный вывод многополюсного компонента �
�
Iz(<имя>) �
IA(T3) �
Ток через входные или выходные зажимы линии передачи �
�



В спецификации переменных V(<имя>), I(<имя>) параметр <имя> указывает на один из двухполюсных компонентов, имена которых начинаются со следующих букв: C ( конденсатор; D ( диод; E ( ИНУН; F ( ИТУТ; G ( ИТУН; H ( ИНУТ; I ( независимый источник тока; L ( индуктивность; R ( резистор; V ( независимый источник напряжения. 


В спецификациях переменных Vx(<имя>), Vxy(<имя>) и Ix(<имя>) параметр <имя> указывает имя трехполюсного или четырехполюсного компонента (см. разд. 3.2), а x и y ( аббревиатуры их выводов (см. табл. 3.18).





Таблица 3.18





Первая буква имени �
Компонент �
Аббревиатура вывода�
�
B �
Арсенид-галлиевый полевой транзистор�
D (сток) �
�
�
�
G (затвор) �
�
�
�
S (исток) �
�
J �
Полевой транзистор �
D (сток) �
�
�
�
G (затвор) �
�
�
�
S (исток) �
�
M �
MОП-транзистор �
D (сток) �
�
�
�
G (затвор) �
�
�
�
S (исток) �
�
�
�
B (подложка) �
�
Q �
Биполярный транзистор�
C (коллектор) �
�
�
�
B (база) �
�
�
�
E (эмиттер) �
�
�
�
S (подложка) �
�



Например, напряжение коллектор(база транзистора Q3 обозначается как VCB(Q3). 


В спецификациях переменных Vz(<имя>), Iz(<имя>) в качестве параметра <имя> указывается имя линии передачи (начинается с буквы T), а символ z принимает два значения: A ( вход, B ( выход линии передачи. 


2. Режим AC. При выводе результатов расчета частотных характеристик к именам переменных, перечисленным выше, добавляются суффиксы: 





Суффикс �
Пример �
Пояснение �
�
�
V(2,3) �
Модуль �
�
M �
VM(2) �
Модуль �
�
DB �
VDB(R1) �
Модуль в децибелах �
�
P �
VCEP(Q3) �
Фаза в радианах �
�
G �
IAG(T2) �
Групповое запаздывание �
�
R �
IR(VIN) �
Действительная часть �
�
I �
II(R13) �
Мнимая часть �
�



В отличие от режимов DC и TRAN в режиме AC возможен вывод токов не всех компонентов, а только следующих: R ( резисторы; C ( конденсаторы; I ( независимые источники тока; L ( индуктивности; T ( линии передачи; V ( независимые источники напряжения. Для вывода токов других устройств необходимо последовательно с ними включить независимый источник напряжения с нулевой ЭДС и затем указать спецификацию тока этого источника. 


3. Анализ уровня внутреннего шума. При расчете уровня внутреннего шума имена выходных переменных имеют стандартный вид: 


INOISE ( эквивалентный уровень шумового напряжения или тока на входе цепи, равный � EMBED Equation.2  ���; 


ONOISE ( уровень напряжения шума на выходе, равный � EMBED Equation.2  ���;  


DB(INOISE) ( эквивалентный уровень шумового напряжения или тока на входе в децибелах; 


DB(ONOISE) ( уровень напряжения шума на выходе в децибелах. 


В программе Probe корень квадратный из спектральной плотности напряжения и тока внутреннего шума выводится в виде V(INOISE), I(INOISE), V(ONOISE). 





3.3.6. Вспомогательные файлы, определение параметров и функций 





Функции, определяемые пользователем, задаются по директиве 





.FUNC  <имя функции>([<аргумент>*]) <тело> 





Количество аргументов находится в пределах от 1 до 10. Имя функции назначает пользователь, но оно не должно совпадать с именами встроенных функций, например SIN и др. Тело функции, содержащее ее описание, состоит из арифметических выражений и стандартных функций. Если аргументы отсутствуют, круглые скобки ( ) все равно нужны. Определение функции должно в задании на моделирование предшествовать ее использованию. Тело функции можно записывать на нескольких строках, помещая символ “+” в первой позиции строки продолжения. Приведем примеры (см. также [41]):





.FUNC   DR(D)   D/57.296 


.FUNC   E(X)   EXP(X) 


.FUNC   APBX(A,B,X)   A+B*X 


.FUNC  P( )   SIN(0.19634954)





Набор стандартных функций целесообразно записывать в файлы и подключать их к заданию на моделирование директивой .INC.


Глобальные параметры задаются директивами вида 





.PARAM <<имя параметра>=<значение>>* 


.PARAM <<имя параметра>={<выражение> }>* 





Значения параметров могут быть константами или выражениями, содержащими константы или другие параметры. Выражения должны заключаться в фигурные скобки { }.


Введем, например, параметры pi=� EMBED Equation.2  ���, рi2=2� EMBED Equation.2  ���, напряжение источника питания VPOWER=5 и используем один из них при описании емкости конденсатора C1:





.PARAM   pi=3.14159265,  pi2=6.2831853,  VPOWER=5v 


.PARAM   VNUM = {8*pi}


C1  2  0  {1/(pi2*10kHz*5k)} 





Имена вводимых параметров не должны совпадать с именем текущего времени TIME и именами стандартных параметров





GMIN ( минимальная проводимость;


TEMP ( текущая температура;


VT ( температурный потенциал p(n-перехода. 





Директиву .PARAM можно использовать внутри описания макромодели для создания локальных параметров (в Design Center 5 это было запрещено). 


С помощью параметров можно определять все параметры моделей устройств и большинство параметров компонентов и директив, за исключением: 


( температурных коэффициентов резистора TC1, TC2, задаваемых в описании резистора на схеме (в модели резистора это, конечно, допускается); 


( параметров кусочно-линейного источника сигнала PWL; 


( линейных и полиномиальных коэффициентов зависимых источников всех типов E, F, G и H (для этого рекомендуется использовать нелинейные источники). 


Параметры нельзя использовать для именования узлов цепи и задания значений переменных в директивах моделирования .AC, .DC и др.


Директивы .PARAM можно помещать в библиотечные файлы. В процессе поиска параметров сначала просматривается задание на моделирование, а затем все подсоединенные библиотечные файлы.


Файл библиотеки компонентов подключается по директиве 





.LIB ["имя файла библиотеки"] 





В файле библиотеки с указанным именем содержится описание встроенных моделей одного или нескольких компонентов (параметры каждого компонента вводятся по директиве .MODEL) или подсхем (описанных с помощью директив .SUBCKT/.ENDS). В этом же файле могут быть помещены комментарии и обращения к другим директивам .LIB. 


При указании в задании на моделирование имени какого-либо конкретного компонента, модель которого содержится в библиотечном файле, в ОЗУ загружается не весь файл, а только его часть, относящаяся к данному компоненту. 


Приведем примеры: 


.LIB "KT315A.mod" ( подключение файла описания модели транзистора КТ315А; 


.LIB "QRUS.lib" ( подключение файла библиотеки отечественных биполярных транзисторов; 


.LIB "D:\PSPICE\LIB\d.lib" ( подключение библиотеки диодов, находящейся на диске C в подкаталоге LIB каталога PSPICE. 


В том же каталоге, из которого запускается пакет PSpice, может находиться файл каталога библиотек nom.lib, в нем перечислены директивы подключения всех используемых библиотек. Приведем пример этого файла, создаваемого пользователем





.LIB  "D:\PSPICE\LIB\qrus.lib" 


.LIB  "D:\PSPICE\LIB\d.lib" 


.LIB  "D:\PSPICE\LIB\digit.lib" 





В таком случае в задании на моделирование указывается директива  .LIB без параметров. 


Другая особенность связана с переменной среды DOS SIMLIBPATH (разд. 1.4), с помощью которой указывается перечень каталогов, которые просматривает программа PSpice в поисках библиотечных файлов, не указанных в файле nom.lib. Пусть, например, эта переменная установлена по команде 





SET SIMLIBPATH=D:\PSPICE\LIB 





Тогда в задании на моделирование достаточно указать имя файла библиотеки без пути доступа к нему, например





.LIB qrus.lib





При работе c графическим редактором Schematics подключение библиотек производится по команде Analysis/Library and Include Files. Причем опции, помеченные звездочкой “*”, делают выбранный библиотечный файл доступным для всех схем, без звездочки ( только для текущей схемы.


Индексные файлы библиотек. В программе PSpice для ускорения поиска нужной модели в библиотечном файле введены индексные файлы, имеющие расширение .ind. Они создаются автоматически при первом обращении по директиве .LIB к библиотеке, в которой произведены изменения. Обратим внимание на то, что создание индексного файла требует больших затрат машинного времени, поэтому часто изменяемую часть библиотечного файла целесообразно оформить в виде отдельного файла. 


Произвольный файл включается в текст задания на моделирование по директиве 





.INC <"имя файла"> 





Например, 





.INC  "band.cir" 


.INC  "D:\MSIM\diode.mod" 





Допускаются четыре уровня включения, причем включаемые файлы не должны иметь заголовков и директив .END. 


При работе c графическим редактором Schematics подключение произвольных текстовых файлов производится по команде Analysis/Library and Include Files.


Конец задания отмечается директивой 





.END 





В одном файле могут помещаться задания на моделирование нескольких цепей, каждое из которых начинается с заголовка задания и заканчивается этой директивой. 





3.3.7. Прочие директивы 





Параметры и режимы работы программы PSpice устанавливаются с помощью директивы (Options)





.OPTIONS [имя опции]* [<имя опции> = <значение>]* 





Например: 





.OPTIONS   NOPAGE   NODE   RELTOL=1e-4 





Опции перечисляются в любом порядке. Они подразделяются на два вида: 1) опции, имеющие численное значение; 2) опции, не имеющие численного значения (их можно назвать флагами, находящимися в положении “включено” или “выключено”). 


Приведем список флагов (в скобках указаны значения по умолчанию): 


ACCT ( вывод статистики времени выполнения всех видов анализа характеристик цепи и других данных о задании на моделирование (N); 


EXPAND ( включение в описание схемы описания макромодели (N); 


LIBRARY ( включение в описание схемы описания моделей из библиотечных файлов (N); 


LIST ( вывод списка всех компонентов цепи (N); 


NOBIAS ( запрещение вывода в выходной файл значений узловых потенциалов в рабочей точке (N); 


NODE ( печать списка соединений (N); 


NOECHO ( запрещение включения в выходной файл части описания схемы, располагаемой после строки с директивой .OPTIONS (N); 


NOMOD ( запрещение вывода списка параметров моделей (N); 


NOOUTMSG ( подавление передачи в выходной файл  сообщений об ошибках моделирования (N); 


NOPAGE ( запрещение перевода страниц в выходном файле (N); 


NOPRBMSG ( подавление передачи в файл данных для программы Probe сообщений об ошибках моделирования (N); 


NOREUSE ( запрещение автоматического сохранения и восстановления информации о режиме по постоянному току при вариации температуры, статистическом анализе, расчете наихудшего случая и при вариации параметров (N); 


OPTS ( вывод значений всех опций (N). 


Если какой-либо флаг не указан, то по умолчанию устанавливается режим, противоположный описанному выше. 


Опции, имеющие численные значения, представлены в табл. 3.19.





Таблица 3.19.





Имя опции �
Наименование �
Размерность �
Значение по умолчанию �
�
ABSTOL �
Допустимая ошибка расчета токов в режиме TRAN �
А �
10� EMBED Equation.2  ����
�
CHGTOL �
Допустимая ошибка расчета заряда в режиме TRAN �
Кулон �
10� EMBED Equation.2  ����
�
CPTIME �
Максимальное время работы процессора, разрешенное для выполнения данного задания транзистора �
с �
10� EMBED Equation.2  ����
�
DEFAD �
Диффузионная площадь стока МОП-транзистора �
м� EMBED Equation.2  ��� �
0�
�
DEFAS �
Диффузионная площадь истока МОП-транзистора �
м� EMBED Equation.2  ��� �
0�
�
DEFL �
Длина канала МОП-транзистора  �
м �
10� EMBED Equation.2  ����
�
DEFW �
Ширина канала МОП- транзистора �
м �
10� EMBED Equation.2  ��� �
�
DIGERRDEFAULT �
Максимальное количество контролируемых ошибок цифровых устройств�
�
20 �
�
DIGERRLIMIT �
Максимальное количество сообщений об ошибках в цифровых устройствах�
�
0 �
�
DIGDRVF�
Минимальное выходное сопротивление цифровых устройств (для моделей UIO)�
Ом�
2�
�
DIGDRVZ �
Максимальное выходное сопротивление цифровых устройств (для моделей UIO) �
кОм �
20�
�
DIGFREQ �
Частота дискретизации при анализе цифровых устройств �
Гц �
109�
�
DIGINITSTATE �
Установка начального состояния триггеров: 0 ( сброс; 1 (  установка; 2 ( X�
�
2 �
�
DIGOVRDRV�
Отношение выходных сопротивлений цифровых устройств, при которых изменяется состояние общего выходного узла�
 �
3�
�
DIGIOLVL �
Уровень интерфейса А/Ц, Ц/А по умолчанию �
�
1�
�
DIGMNTYMX �
Селектор выбора задержки цифрового устройства по умолчанию: 1 ( минимум; 2 ( типичное значение; 3 ( максимум; 4 ( мин/макс (наихудший случай для цифровых устройств) �
�
2�
�
DIGMNTYSCALE �
Масштабный коэффициент для расчета минимальной задержки �
�
0,4�
�
DIGTYMXSCALE �
Масштабный коэффициент для расчета максимальной задержки �
�
1,6�
�
DIGOVRDRV �
Отношение выходных сопротивлений цифровых устройств, при которых изменяется состояние общего выходного узла�
�
3�
�
DISTRIBUTION �
Закон распределения отклонений параметров от номинальных значений �
�
UNIFORM�
�
GMIN �
Минимальная проводимость ветви цепи (проводимость ветви, меньшая GMIN, считается равной нулю)  �
См �
10� EMBED Equation.2  ����
�
ITL1 �
Максимальное количество итераций в режиме DC �
�
40  �
�
ITL2 �
Максимальное количество итераций при расчете передаточных функций по постоянному току при переходе к последующей точке �
�
20�
�
ITL4 �
Максимальное количество итераций при переходе к следующему моменту времени в режиме TRAN�
�
10  �
�
ITL5 �
Общее максимальное количество всех итераций в режиме TRAN (установка ITL5=0 означает бесконечность)�
�
0�
�
LIMPTS �
Максимальное количество точек, выводимых в таблицу или на график (LIMPTS=0  означает бесконечность) �
�
0  �
�
NUMDGT �
Количество значащих цифр в таблицах выходных данных (не более 8)�
�
4�
�
PIVREL �
Относительная величина элемента строки матрицы, необходимая для его выделения в качестве ведущего элемента (режим AC) �
�
10� EMBED Equation.2  ����
�
PIVTOL �
Абсолютная величина элемента строки матрицы, необходимая для его выделения в качестве ведущего элемента (режим AC) �
�
10� EMBED Equation.2  ����
�
RELTOL �
Допустимая относительная ошибка расчета напряжений и токов в режиме TRAN �
�
0,001�
�
TNOM �
Номинальная температура �
� EMBED Equation.2  ���С �
27 �
�
VNTOL �
Допустимая ошибка расчета напряжений в режиме TRAN �
В �
10� EMBED Equation.2  ����
�
WIDTH �
Длина строки выходного файла (аналогично директиве .WIDTH) �
�
80�
�



В процессе моделирования программа PSpice генерирует различные сообщения, которые передаются в выходной файл и файл данных для программы Probe.  Список сообщений о специфических ошибках в процессе моделирования цифровых устройств приведен в табл. 3.20.





Таблица 3.20





Тип сообщения�
Объяснение ошибок�
�
Нарушение временных соотношений�
�
SETUP�
Минимальное время установки сигнала�
�
HOLD�
Минимальное время удержания сигнала�
�
RELEASE�
Минимальное время, в течение которого сигнал неактивный (обычно для выполнения контроля типа CLEAR)�
�
WIDTH�
Минимальная длительность импульсов�
�
FREQUENCY�
Минимально или максимально допустимая частота переключений�
�
GENERAL�
Обнаружение ошибки устройством контроля, заданного с помощью булевского выражения�
�
Гонки сигналов�
�
AMBIGUILTY CONVERGENCE�
Временная неопределенность, связанная с одновременным поступлением на один вход импульсов со спадающим и нарастающим фронтами�
�
CUMULATIVE AMBIGUILTY �
Неопределенности уровней сигналов, связанные с суммированием частично перекрывающихся фронтов импульсов�
�
DIGITAL INPUT VOLTAGE�
Предупреждающее сообщение о том, что входной сигнал находится вне допустимого диапазона напряжений. Моделирование продолжается, заменяя входное напряжение граничным значением�
�
NER-STATE CONFLICT�
При подаче на один вход одновременно нескольких сигналов различного уровня моделировщик заменяет его неопределенным сигналом Х �
�
SUPPRESSED GLITCH�
Короткие импульсы, длительность которых меньше задержки переключения, игнорируются моделировщиком�
�
PERSISTENT HAZARD�
Появление гонок сигналов на внешних портах �
�
ZERO-DELAY-OSCILLATION�
При изменении логического уровня выходного узла более чем 50 раз за единичный временной дискрет узел считается осциллирующим и моделирование прекращается�
�



Статистические сведения о задании выводятся в выходной файл с расширением .out при введении опции ACCT в директиве .OPTION. Перечень выводимых данных приведен в табл. 3.21.





Таблица 3.21 





Параметр �
Значение �
�
NUNODS �
Количество узлов схемы устройства без учета подсхем �
�
NCNODS �
Количество узлов схемы устройства с учетом подсхем �
�
NUMNOD �
Общее количество узлов схемы замещения устройства с учетом внутренних узлов встроенных моделей компонентов �
�
NUMEL �
Общее количество компонентов устройства с учетом подсхем �
�
DIODES �
Количество диодов с учетом подсхем �
�
BJTS �
Количество биполярных транзисторов с учетом подсхем �
�
JFETS �
Количество полевых транзисторов с учетом подсхем �
�
MFETS �
Количество МОП-транзисторов с учетом подсхем �
�
GASFETS �
Количество арсенид-галлиевых полевых транзисторов с учетом подсхем �
�
NDIGITAL �
Количество цифровых устройств с учетом подсхем �
�
NSTOP �
Размерность воображаемой матрицы цепи (фактически не все элементы разреженных матриц хранятся в памяти) �
�
NTTAR �
Фактическое количество входов в матрице цепи в начале вычислений �
�
NTTBR �
Фактическое количество входов в матрице цепи в конце вычислений �
�
NTTOV �
Количество ненулевых элементов матрицы цепи �
�
IFILL �
Разность между NTTAR и NTTBR �
�
IOPS �
Количество операций с плавающей запятой, необходимых для решения одного матричного уравнения цепи �
�
PERSPA �
Степень разреженности матрицы цепи в процентах �
�
NUMTTP �
Количество шагов интегрирования переходного процесса �
�
NUMRTP �
Количество моментов времени при расчете переходного процесса, при которых шаг интегрирования был слишком велик и расчет повторен с меньшим шагом �
�
NUMNIT �
Общее количество итераций при расчете переходного процесса �
�
DIGTP �
Количество временных шагов при логическом моделировании �
�
DIGEVT �
Количество логических событий �
�
DIGEVL �
Количество вычислений логических состояний цифровых устройств �
�
MEMUSE �
Размер используемой области ОЗУ в байтах �
�
READIN �
Время, затраченное на чтение входного файла и поиск ошибок в нем �
�
SETUP �
Время, затраченное на формирование матрицы цепи �
�
DC sweep �
Время, затраченное на расчет передаточных функций по постоянному току �
�
Bias point�
Время, затраченное на расчет режима по постоянному току в рабочей точке �
�
Matrix solution�
Время, затраченное на решение матричного уравнения�
�
Matrix load�
Время, затраченное на составление уравнений компонентов�
�
Digital simulation�
Время, затраченное на вычисление логических состояний цифровых устройств �
�
AC and noise�
Время, затраченное на расчет в частотной области�
�
Transient analysis �
Время, затраченное на расчет переходного процесса�
�
Monte Carlo�
Время, затраченное на выполнение директив .MC и .WCASE�
�
OUTPUT �
Время, затраченное на переформатирование данных, необходимое перед выполнением директив .PRINT и .PLOT �
�
OVERHEAD �
Время, затраченное на выполнение задания �
�
Total job time�
Общее время выполнения задания, за исключением времени, затраченного на загрузку файлов программы PSpice �
�



В начале знакомства с PSpice рекомендуется воспользоваться установкой параметров директивы .OPTIONS по умолчанию и изменять их по мере надобности после приобретения опыта моделирования. 


Задание текстовых переменных, текстовых выражений и имен файлов производится по директиве 





.TEXT <<имя> = "<имя файла>">* 


.TEXT <<имя> = |<текстовое выражение>|>* 





Директива .TEXT используется только при моделировании цифровых устройств. Текстовые параметры используются в следующих случаях: 


( для указания имени JEDEC-файла программируемых логических матриц;


( для указания Intel Hex-файла для программирования ПЗУ или начальной записи в ОЗУ; 


( для указания имени файла генератора цифровых сигналов FSTIM; 


( для задания текстового параметра макромодели; 


( как часть текстового выражения, используемого в указанных выше целях. 


Спецификация внешних портов выполняется по директиве (введена впервые в Design Center 6.1)





.EXTERNAL   <атрибут>   <имя узла>*





Параметр <атрибут> принимает одно из значений INPUT (входной), OUTPUT (выходной) или BIDIRECTIONAL  (двунаправленный) в соответствии с назначением порта. Внешние порты предназначены для подсоединения периферийных устройств (например, генераторов сигналов) к моделируемой схеме. Приведем примеры





.EXTERNAL    INPUT   Data1,   Data2,   Data3


.EXTERNAL    OUTPUT   P1


.EXTERNAL    BIDIRECTIONAL   BPort1   BPort2   BPort3





Задание имени файла с описанием внешних воздействий выполняется по директиве (введена впервые в Design Center 6.1)





.STIMLIB   <имя файла[.stl]>





Файл описания внешних сигналов с расширением .stl создается программой Stimulus Editor (разд. 7.1) и делается доступным для моделирования с помощью этой директивы. Расширение имени файла, в том числе и .stl, указывать обязательно.


Задание внешних воздействий выполняется с помощью директивы (введена впервые в Design Center 6.1)





.STIMULUS    <имя сигнала>      <тип>   <параметры>*





Приведем примеры





.STIMULUS   InputPulse  PULSE (-1mv  1mv  2ns  2ns  50ns  100ns)


.STIMULUS   DigitalPulse   STIM  (1,1)


+	0s	1


+	10ns	0


+	20ns	1


.STIMULUS   50KHZSIN   SIN   (0   5   50kHz   0   0   0)





Директива .STIMULUS вносится в библиотеки воздействий, созданных с помощью программы Stimulus Editor. Каждое воздействие имеет имя <имя сигнала>, присваиваемое источникам напряжения или тока (типа V или I) или источникам цифровых сигналов (типа STIM).


3.4. Примеры моделирования аналоговых устройств 





3.4.1. Транзисторный полосовой усилитель 





�


Рис. 3.20. Принципиальная схема полосового усилителя


На рис. 3.20 приведена принципиальная схема транзисторного полосового усилителя, на примере которой продемонстрируем основные возможности программы PSpice. Усилитель собран на биполярном транзисторе КТ315А, который нагружен на двухконтурный полосовой LC-фильтр. Этот фильтр настроен на частоту 36 МГц и имеет характеристическое сопротивление 100 Ом. Фильтр состоит из двух параллельных колебательных контуров L1(C4, L2(C6 с емкостной связью C5. Связь фильтра с транзистором непосредственная через разделительный конденсатор C3, а с нагрузкой RLOAD ( с помощью согласующего высокочастотного трансформатора L2(L3, обмотки которого имеют коэффициент связи 0,8. Резисторы RK1, RK2 имитируют потери в контурах (в реальной схеме усилителя они, естественно, отсутствуют). Усилитель питается от источника постоянного напряжения 6 В. Ко входу должен быть подключен источник гармонического сигнала. 


Задание на моделирование занесено в файл band.cir : 





BANDPASS FILTER 


*ХAРAКТЕРИСТИКИ LC-ФИЛЬТРA 


.OPTIONS   ACCT   LIST   NODE   OPTS   NOPAGE   RELTOL=1e-4 


RG   	IN 	 10  	1k


R1   	 2    	 1 	MRES  4.5k


R2  	 1     	 0 	MRES 1.3k 


R3   	2	 3 	100 


R4 	4	 0	MRES 200 


RK1	 5 	0 	2.31k


RK2 	6	0	2.31k 


RLOAD 	9 	0 	100 


C1 	IN 	1	 1000pF 


C2 	4 	0 	3300pF 


C3 	5 	3 	50nF 


C4 	5 	0 	241.4pF


C5 	6 	5 	44.5pF 


C6 	6 	0 	243pF 


L1 	5 	0 	0.0692u 


L2 	6 	0 	0.0692u 


L3 	0 	9 	0.108u


K1 	L2 	L3 	0.8 


Q1 	3 	1 	4 	KT315A 


VS 	2 	0 	DC 	6V 


VIN 	10 	0 	AC 	1 	SIN ( 0  0.05V 36MEG ) 


.MODEL   MRES   RES ( R=1 DEV=20% TC1=0.01 ) 


.LIB    bip_rus.lib 


.TEMP    -37   27    67 


.DC    VS    0.5    20.5    5 


.TF     V(3)   VS 


.SENS   VBE(Q1) 


.AC    LIN    41   16MEG   56MEG 


.NOISE   V(9)   VIN 


.MC    5    AC    V(9)    YMAX 


.TRAN/OP    5ns    1us 


.FOUR    36MEG    V(9) 


.PRINT    NOISE    INOISE    ONOISE 


.PROBE    V(3)    V(6)    V(9)    V(INOISE) 


.END 





В первых двух строках этого файла указан заголовок задания и дан комментарий. На следующей строке по директиве .OPTIONS  устанавливаются параметры выдачи результатов моделирования в выходной файл band.out и допустимая погрешность вычислений. Затем идет описание схемы усилителя (порядок следования строк в этом файле совершенно произвольный). На вход подключен источник напряжения VIN, ЭДС которого в режиме AC равна 1 В (в этом режиме рассчитываются частотные характеристики линеаризованной цепи, поэтому абсолютное значение ЭДС входного сигнала не принципиально), а в режиме TRAN имеет вид гармонического сигнала с частотой 36 МГц и амплитудой 0,05 В (здесь уже ЭДС входного сигнала имеет реальную амплитуду, так как переходные процессы рассчитываются для исходной нелинейной цепи). Резисторы R1, R2 и R4 описываются по директиве .MODEL моделью с именем MRES, параметры которой перечислены в круглых скобках (они имеют линейный температурный коэффициент сопротивления TC1=0,01, разброс сопротивлений имеет относительное среднеквадратическое отклонение DEV=20 %). Директива .LIB указывает, что параметры модели биполярного транзистора находятся в файле bip_rus.lib. По директиве .TEMP устанавливается список температур, при которых будет проводиться анализ характеристик усилителя. По директиве .DC определяется режим расчета по постоянному току при вариации напряжения источника питания VS от 0,5 до 20,5 В с шагом 5 В. Малосигнальная передаточная функция dV(9)/dVIN рассчитывается по директиве .TF. Далее по директиве .SENS рассчитывается чувствительность постоянного напряжения база(эмиттер транзистора VBE(Q1) к изменению параметров всех компонентов схемы. По директивам .AC и .NOISE производится расчет частотных характеристик и уровня шума в диапазоне частот от 16 до 56 МГц (всего 41 точка по частоте). После этого по директиве .TRAN рассчитываются переходные процессы в интервале времени от 0 до 1 мкс (шаг интегрирования устанавливается автоматически) с шагом вывода данных в файл band.out, равным 5 нс. После этого директива .FOUR задает спектральный анализ по Фурье напряжения на нагрузке V(9), частота первой гармоники установлена равной 36 МГц. Директива .MC задает статистический анализ в режиме AC модуля напряжения V(9); всего проводится пять статистических испытаний при случайных значениях сопротивлений R1, R2 и R4, которые имеют ненулевое значение отклонения DEV. 


Перечень переменных, выводимых в файл band.out в виде таблиц, приводится в директиве .PRINT. Этот файл по окончании работы программы PSpice просматривается, редактируется и выводится на печать средствами DOS (при этом графики изображаются алфавитно-цифровыми символами). Графические возможности ПК реализуются с помощью графического постпроцессора Probe. В директиве .PROBE перечислен список переменных,  заносимых в файл результатов band.dat, который автоматически загружается в программу Probe после ее вызова (см. разд. 7.2). В качестве примера на рис. 3.21 показаны два графика семейства АЧХ напряжений V(9) в узле 3 и корня квадратного из спектральной плотности напряжения эквивалентного шума на входе INOISE при вариации температуры. Фрагмент выходного файла band.out приведен в Приложении 3. 


�Рис. 3.21. Семейство АЧХ выходного напряжения и уровня входного шума при вариации температуры











3.4.2. Генератор Ван-дер-Поля 





Рассмотрим методику моделирования динамических систем, заданных дифференциальными уравнениями, на примере генератора Ван-дер-Поля: 





� EMBED Equation.2  ���� EMBED Equation.2  ���� EMBED Equation.2  ���


Для применения программы PSpice к решению этого уравнения синтезируем электрическую цепь, подчиняющуюся этому же уравнению. Эта задача решается неоднозначно. Проще всего установить аналогию с параллельным колебательным LC-контуром, к которому подключен нелинейный управляемый источник тока G (рис. 3.22). Выберем в качестве переменной x ток через индуктивность � EMBED Equation.2  ���. Тогда производная dx/dt определяется напряжением на индуктивности 


� EMBED Equation.2  ���. 


Уравнение такого контура относительно переменной x имеет вид


� EMBED Equation.2  ���


�Рис. 3.22. Модель генератора Ван-дер-Поля





Отсюда вытекают условия эквивалентности этого уравнения уравнению генератора Ван-дер-Поля





� EMBED Equation.2  ���   � EMBED Equation.2  ���





Последовательно с индуктивностью L на рис. 3.22 включен источник напряжения VL с нулевой ЭДС, чтобы иметь возможность током через него управлять источником тока G (по правилам PSpice допускается управление только разностью потенциалов любых узлов цепи и токами через независимые источники напряжения). Приведем задание на моделирование уравнения Ван-дер-Поля: 





Van-der-Pole Oscillator 


.OPTIONS RELTOL=1e(4 


.PARAM EPS=1 


L   2    0  1   IC=0.2A 


C   1    0  1 


VL   1  2   DC=0 


G    0   1   VALUE={EPS*(1(I(VL)*I(VL))*V(1)} 


.TRAN   1ms   100s  SKIPBP 


.WATCH   TRAN   I(VL)   V(1) 


.PROBE 


.END 





�Рис. 3.23. Фазовый портрет генератора


Обратим внимание, что в директиве .TRAN ключевое слово SKIPBP отменяет расчет режима по постоянному току. Начальное значение тока через индуктивность установлено равным IL(0)=0,2 A. Еще одна особенность задания ( директива .WATCH, с помощью которой в процессе расчета переходного процесса на экран программы PSpice выводятся текущие значения переменных I(VL) и V(1), т. е. x и dx/dt. По результатам моделирования с помощью программы Probe построен фазовый портрет генератора ( зависимость производной dx/dt от x (рис. 3.23). 








3.4.3. Следящая система 





Программа PSpice разработана для моделирования нелинейных электрических цепей, заданных принципиальными электрическими схемами. Но это не препятствует моделированию систем автоматики, заданных функциональными схемами. Необходимо лишь каждому элементу функциональной схемы поставить в соответствие схему замещения из компонентов, допустимых в PSpice. Наиболее удобно это делать с помощью линейных и нелинейных управляемых источников (см. 3.2.4, 3.2.5). 


Проиллюстрируем эту методику на примере следящей системы с отрицательной обратной связью, показанной на рис. 3.24, а, где передаточная функция � EMBED Equation.2  ���.  Моделирование проведем двумя способами.


1. Сначала с помощью старой версии PSpice все три блока функциональной схемы ( сумматор, усилитель с коэффициентом передачи K и фильтр с передаточной функцией G(s) ( реализуем с помощью линейных источников напряжения EE, EK, EZ, управляемых напряжениями, как показано на рис. 3.24, б. Источник сигнала имитируется независимым источником напряжения VX. Параллельно каждому источнику включены сопротивления RX, RE, RK, RZ, чтобы в схеме замещения не было разомкнутых контуров (это требование PSpice, связанное с ошибками в программе контроля топологии). Потенциалы узлов в схеме на рис. 3.24, б соответствуют значениям переменных в одноименных точках функциональной схемы на рис. 3.24, а. Приведем задание на расчет частотных характеристик этой системы: 





СЛЕДЯЩАЯ СИСТЕМА С ТРЕХЗВЕННОЙ RC-ЦЕПЬЮ 


.PARAM  K=1  A=20  TAU=0.1 


VX  1  0  AC=1 


RX  1  0  1 


*EE  0  2  VALUE={V(1)-V(4)} 	; ОС замкнута 


EE  0  2  VALUE={V(1)} 		; ОС разомкнута 


RE  2  0  1 


EK  0  3  VALUE={K*V(2)} 


RK  3  0  1 


EL  4  0  LAPLACE {V(3)} {A/((1+TAU*s)*(1+TAU*s)*(1+TAU*s))} 


RL  4  0  1 


.AC  LIN  500  0.01 10 


.WATCH  AC  V(2)  V(3)  V(4) 


.PROBE 


.END 


�


�


Рис. 3.24.  Функциональная схема следящей системы (а), ее модель для старой версии PSpice (б) и графическое изображение в Schematics (в)





Здесь предусмотрено два варианта сумматора EE ( для моделирования следящей системы с замкнутой и разомкнутой обратной связью. Последнее используется для исследования устойчивости следящей системы с помощью критерия Найквиста. Годограф коэффициента передачи разомкнутой системы для иллюстрации показан на рис. 3.25: точка на комплексной плоскости с координатами ((1, j0) годографом не охватывается, поэтому рассматриваемая система устойчива. 


�Рис. 3.25. Годограф коэффициента передачи разомкнутой следящей системы


2. В Design Center 6.2 больше не нужно подсоединять фиктивные резисторы к  управляющим входам функциональных источников ( в результате функциональная схема следящей системы (рис. 3.24, а), нарисованная в Sсhematics, имеет естественный вид (рис. 3.24, в). В то же время текстовое задание на моделирование, помещающееся теперь в трех файлах с расширениями .cir, .net и .als, изменилось незначительно:





Файл control.cir





* C:\MSIM62J\EXAMPLES\CONTROL.SCH


* Schematics Version 6.2j - December 1995


* Fri Feb 02 16:17:26 1996


.PARAM    A=20   TAU=0.1   K=1


** Analysis setup **


.ac   LIN   501   0.01   10


* From [SCHEMATICS NETLIST] section of msim.ini:


.lib   C:\MSIM62J\LIB\MY.LIB


.lib   nom.lib


.INC  "CONTROL.net"


.INC  "CONTROL.als"


.probe


.END





Файл control.net





* Schematics Netlist *


V_V1         1    0   DC   0V   AC   1V 


E_DIFF1   2   0   VALUE {V(1,4)}


E_GAIN1  3   0   VALUE {{K} * V(2)}


E_LAPLACE1    4   0   LAPLACE {V(3)} {(A)/((1 + s*TAU)*


+	(1+s*TAU)*(1+s*TAU))}





Файл control.als





* Schematics Aliases *


.ALIASES


V_V1                V1(+=1 -=0 )


E_DIFF1          DIFF1(OUT=2 IN2=1 IN1=4 )


E_GAIN1          GAIN1(OUT=3 IN=2 )


E_LAPLACE1          LAPLACE1(OUT=4 IN=3 )


_    _(1=1)


_    _(2=2)


_    _(4=4)


_    _(3=3)


.ENDALIASES





В заключение отметим, что разнообразным применениям программы PSpice посвящены работы [35, 49, 55]. В статье [36] обсуждаются специфические вопросы исследования переходных процессов в коммутируемых цепях. Практические советы по моделированию регулярно помещаются в журнале “The Design Center Source”, выпускаемом корпорацией MicroSim.

